

クォータニオンにはマトリクスがよく似合う (増補)

SHIMURA Masato
jcd02773@nifty.ne.jp

2023年8月31日

目次

1	手計算の Example を探す	2
2	クォータニオンのアルジェブラ	3
3	掛け算の段	6
4	虚数 3 人娘早変わりの段	9
5	回転のアルジェブラ	10
6	マトリクスを用いた掛け算	14
7	公道走行	17
8	Conclusion	19

はじめに

ハミルトンは 3 元数が欲しかったがスカラの ω が出てきて 4 元数になってしまった。

「読んで分からないのは書いた人が良く分かっていないからだ」明快な助言を頂いて、買い込んであった本を閉じ、Net 上の和文英文の文献を探し求めた。

クォータニオンをツールでなく言語ベース使い込んだ若手エンジニアの手稿が Quiita に幾つか上がるようになってきた。

英語の講義用の幾つかのプリントはツール抜き、例題付きで説明も丁重で論文に在りがちな難解な数式もなく、理解とスクリプトの確認に大いに役立った。

スクリプトを組んで実感したことだが、クォータニオンの掛け算と回転はマトリクスとよく馴染み、クォータニオンのマトリクスをくみ上げれば内積計算が全てやってくれる。

本稿は講義プリントからクォータニオン数学の式を抜き出すことから始め、スクリプトを組み上げてマトリクス方式にたどり着くまでの記録であり、冗長なところもあるがメモとし

て残した。

J のアドオンの math/mt にクォータニオンが入っている。例によって説明がないので手探りテスト。

クォータニオン

ハミルトンがダブリンの橋桁に閃いたクォータニオンの式を書き付けたのは 1843 年。時間 (t) でない 4 次元の ω は不思議な存在である。

この ω のグラフィックスのイメージが最後までうまくつかめなかったが「鶉匠と繋がれた 3 羽の虚数鶉」, 「3 人の虚数海女と船上の一人の漁夫」と考えることにした。

$j = w$ は実数で回転を表す。計算はラジアンで行う。J の numeric.ijs に rfd と dfr が入っておりラジアン、デグリーの相互計算ができる。x, y, z は虚数= i

$$\omega + ix + jy + kz$$

スカラー 1 個と虚数 3 個の書式はローカル色もあるが概ね上が標準。

1 手計算の Example を探す

今や各種教科書でも絶滅危惧種になった手計算の例をネットで探すと次が見つかった。何れも回転の例題である。

1.1 高校数学の美しい物語

点 A:(3,0,0) を $y = x, z = 0$ で表される直線の回りに 180° 回転させた点 B の座標を求めよ。

- 点 A に対応する四元数は $a = 3i$
- 回転 (軸) $u = (\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}, 0)$ なので, 対応する四元数は $q = \frac{1}{\sqrt{2}}(i + j)$
- $b = \frac{1}{\sqrt{2}}(i + j)(3i)\frac{1}{\sqrt{2}}(-i - j) = \dots = 3j$
- B の座標は (0,3,0) である。

Memo

1. 回転軸 (3D ベクトル) と回転角
2. ノルム

% %:2

0.707107

$$\frac{1}{\sqrt{2}}$$

1.2 物理のカギしっぽ

ベクトル (0,0,1) を
y 軸の回りに 90 度回転さ
せる

回転させるポイント: $x = (0, 0, 1)$ の四元数表示

$$x = k$$

回転軸と回転角:y 軸の回りでの 90 度回転の四元数表現

$$q = \cos \frac{\pi}{4} + j \sin \frac{\pi}{4} = \frac{1}{\sqrt{2}} + j \frac{1}{\sqrt{2}}$$

計算:

$$x' = q_x \bar{q}$$

$$\begin{aligned} &= \left(\frac{1}{\sqrt{2}} + j \frac{1}{\sqrt{2}}\right) k \left(\frac{1}{\sqrt{2}} - j \frac{1}{\sqrt{2}}\right) \\ &= \left(\frac{1}{\sqrt{2}} + j \frac{1}{\sqrt{2}}\right) \left(k \frac{1}{\sqrt{2}} + i \frac{1}{\sqrt{2}}\right) \\ &= k \frac{1}{2} + i \frac{1}{2} + i \frac{1}{2} - k \frac{1}{2} \\ &= i \end{aligned}$$

クォータニオンの $x' = i$ はベクトル (1,0,0) のこと

2 クォータニオンのアルジェブラ

クォータニオンは複素数を拡張した数であり、数に関するルールはおおむね適用される。
ハミルトンの四元数の特性は、非群的/非線形的であり、 $qp \neq pq$ で非可換/Non-abelian
である。

(幾つかの文献の抜き書きを編集したので記号の不一致がある。)

最初に J で `require 'plot numeric trig'` と打って、円関数と数値計算の小物ツールをロードしておく

2.1 ポーズの段

$$q = q_0 + \mathbf{q} = q_0 + q_1i + q_2j + q_3k$$

パートナー

$$p = p_0 + \mathbf{p} = p_0 + p_1i + p_2j + p_3k$$

Example

$$\omega + 4i + 4j - 4k$$

2.2 足し算の段

$$p + q = (p_0 + q_0) + (p_1 + q_1)i + (p_2 + q_2)j + (p_3 + q_3)k$$

2.3 共役/conjugate の段

1. 数式

$$q^* = q_0 - \mathbf{q} = q_0 - q_1i - q_2j - q_3k$$

2. 実行

```
conjugate0 1 4 4 _4
1 _4 _4 4
```

3. こちらは正規化してある

```
conjugateQ 1 4 4 _4
0.142857 _0.571429 _0.571429 0.571429
```

```
+/*: conjugateQ 1 4 4 _4
1
```

4. Script

```
conjugate0=: 3 : '({.y ),- }. y'
conjugateQ=: conjugate0@unitQ
```

2.4 ノルムの段

1. 数式

$$\|q\| = \sqrt{qq^*} = \sqrt{q^*q} = \sqrt{q_0^2 + q_1^2i + q_2^2j + q_3^2k}$$

2. 手計算

$$|q| = \sqrt{1 + 16 + 16 + 16} = \sqrt{49} = 7$$

3. 実行

```
norm 1 4 4 _4
7
```

4. Script

```
norm=:[: %: [: +/ *:
```

5. また $|q| = q^*q = qq^*$

```
] a0=. 1 4 4 _4,. conjugateQ 1 4 4 _4
1 0.142857
4 _0.571429
4 _0.571429
_4 0.571429
```

```
multipleQ a0
7 0 0 0
```

6. 次の書式もある

$$|q|^2 = \omega^2 + x^2 + y^2 + z^2$$

2.5 normalization/ユニット化/正規化の段

$$Uq = \frac{q}{\|q\|}$$

$$q' = \frac{q}{|q|} = \frac{1 + 4i + 4j - 4k}{7} = \frac{1}{7} + \frac{4}{7}i + \frac{4}{7}j - \frac{4}{7}k$$

```
unitQ 1 4 4 _4
```

```
0.142857 0.571429 0.571429 _0.571429
```

```
+/*: unitQ 1 4 4 _4  
1
```

```
unitQ=:3 : 0  
NB.usage: unitQ 1 1 1 // unitQ 1 1 1 1  
if. 4= # y do. y % norm y  
else. 0, y % norm y NB. make unit Q  
end.  
)
```

ベクトルとクォータニオンをカバーしている。

2.6 逆数/inverse の段

$$q^{-1} = \frac{q^*}{\|p\|}$$

規格化されているときは $\text{inverse } q^{-1}$ は $\text{conjugate } q^*$ で足りる

$$\text{inverse } q^{-1} = \frac{q^*}{|q|^2}$$

```
inverseQ=: 3 : '( conjugateQ y) % norm y'
```

このスクリプトは $q^{-1} = \frac{q^*}{|q|}$

英語の資料まで目を通すとノルムの定義 2 通りあるが、ここでは多数派のルート付きの方である。

- $\sqrt{\omega^2 + xi^2 + yj^2 + kz^2}$
- $\omega^2 + xi^2 + yj^2 + kz^2$

1 が ω に出てくる

```
] a4=. 1 4 4 _4 ,.inverseQ 1 4 4 _4  
1 0.0204082  
4 _0.0816327  
4 _0.0816327  
_4 0.0816327
```

```
clean multipleQ_mat a4  
1 0 0 0
```

3 掛け算の段

ここで扱うクォータニオンとクォータニオンの掛け算はハイライトであるが、複雑なので一節を設けよう

3.1 スカラー倍

スカラー倍は nq

3.2 入力のフォーマット

先に入力フォーマットを確認しておこう。クォータニオンを作成して掛け算や回転を行う場合は原則として次を用いる。

左の q はクォータニオン、 p はクォータニオン化したベクトル

```
qp
3 2
1 _1
_2 2
1 3
```

3.3 クォータニオン同士の掛け算—チキチキバンバン方式

-

$$pq = p_0q_0 - \mathbf{p} \cdot \mathbf{q} + p_0\mathbf{q} + q_0\mathbf{p} + \mathbf{p} \times \mathbf{q}$$

1. $\mathbf{p} \cdot \mathbf{q}$ は内積
 2. $\mathbf{p} \times \mathbf{q}$ は外積である
 3. (弓) (内積) (射撃—バンバンバン) (外積) ... チキチキバンバン方式
- これは次の結果である

$$[p_0\mathbf{p}][q_0\mathbf{q}] = p_0q_0 - \mathbf{p} \cdot \mathbf{q} + p_0\mathbf{q} + q_0\mathbf{p} + \mathbf{p} \times \mathbf{q}$$

- 経過を書き下してみよう

$$\begin{aligned}
pq &= (p_0 + p_1i + p_2j + p_3k)(q_0 + q_1i + q_2j + q_3k) \\
&= (p_0q_0) \\
&\quad - (p_1q_1 + p_2q_2 + p_3q_3) \\
&\quad + p_0(q_1i + q_2j + q_3k) \\
&\quad + q_0(p_1i + p_2j + p_3k) \\
&\quad + (p_2q_3 - p_3q_2)i + (p_3q_1 - p_1q_3)j + (p_1q_2 - p_2q_1)k
\end{aligned}$$

目を凝らすと、同じ虚数単位の積は内積に、異なる虚数単位の積は外積になる

- 例題で経過を確認する

1. 問

$ \begin{aligned} q &= 3 + i - 2j + k \\ p &= 2 - i + 2j + 3k \end{aligned} $

2. 内積

$$\begin{aligned}
&1 \quad 2 \quad 1 \quad + / \quad . \quad * \quad 1 \quad 2 \quad 3 \\
&-2
\end{aligned}$$

3. 外積

$$\begin{aligned}
&\text{vec_outer } 1 \quad 2 \quad 1, \dots 1 \quad 2 \quad 3 \\
&_8 \quad _4 \quad 0 \\
&-8i - 4j
\end{aligned}$$

4. 構成

$$\begin{aligned}
pq &= 6 - (-2) \\
&3(-i + 2j + 3k) \\
&2(i - 2j + k) \\
&-8i - 4j + 0k \\
\hline
&8 - 9i - 2j + 11k
\end{aligned}$$

5. 外積のスクリプト

```

det=: -/ . *      NB. determinant
index=:  }. L:0  ({@>0 1 2) |. (L:0)  0 1 2
vec_outer=: 3 : ' ; det L:0  index { L:0  y'

```

Jでアルジェブラのスクリプトも作っておこう。

1. multiple

空を飛ぶ魔法の車チキチキバンバン (ChittyChittyBangBang) 式。歌に乗って楽しく!
(YouTubeにあるよ)


```

qp
3 2
1 _1
_2 2
1 3

multipleQ qp
8 _9 _2 11

```

```

multipleQ=: 3 : 0 NB. OK
NB. chiqichiqui-bangbang metod
NB. Usage: multipleQ qp
qp=: y
a0=: */ {. qp
v0=: |: }. qp NB. take & rotate vector part
a1=: ({. v0) +/ . * {: v0
a2=: +/ ({.qp) * |.v0
a3=: vec_outer |: v0
(a0-a1),a2 - a3 NB. po+P
)

```

2. クォータニオンでは $qp \neq pq$ である。

```

qp
3 2
1 _1
_2 2
1 3

multipleQ qp
8 _9 _2 11
multipleQ pq=. |. "1 qp
8 7 6 11

```

4 虚数 3 人娘早変わりの段

$$i^2 = j^2 = k^2 = ijk = -1$$

$$ij = k = -ji$$

$$jk = i = -kj$$

$$ki = j = -ik$$

この部分は原理確認には重要だがクォータニオンが計算過程で自動的にやってくれる。

- $i \cdot j = k$

```

] a=. 0 1 0 0 ,. 0 0 1 0
0 0
1 0
0 1
0 0

```

- $j = \text{入力形式} = i$
名前は何でもいいが、入力はこの形。 ,. で連結する

```

] a=:qp=: 0 1 0 0 ,. 0 0 1 0
0 0
1 0
0 1
0 0

```

```

multipleQ_mat a
0 0 0 1

```

- $j \cdot i = -k$

```

] a1=. 0 1 0 0 ,. ~ 0 0 1 0
0 0
0 1
1 0
0 0

```

```

multipleQ a1
0 0 0 _1

```

ij が合体すると k に変身し、 xy 軸から z 軸に飛び移る。この時 i, j, k は序列だと考えるとくっついたときに序列が逆転していると変身したときに符号がマイナスになる。

5 回転のアルジェブラ

- $j = \text{入力形式} = i$
入力形式は qp 名前は任意だが ,. で連結する。 w は実数でラジアン、 x, y, z は虚数。

```

qp
0.92388 0
0 2
0 0
0.382683 0

```

q には正規化した上で \cos, \sin の式を入れ込む。 p はそのまま。

- 方向ベクトル $\lambda_x, \lambda_y, \lambda_z$ を軸として

θ だけ回転する

回転クォータニオンは次の通り

$$\left(\cos\frac{\theta}{2}, \lambda_x \sin\frac{\theta}{2}, \lambda_y \sin\frac{\theta}{2}, \lambda_z \sin\frac{\theta}{2}\right)$$

5.1 乗算クォータニオンの作成

回転には (2) 回転クォータニオンと (1) 回転ポイントが必要

1. 回転ポイント

回転ポイントは 3 次ベクトルならばこれに $\omega = 0$ を加えてクォータニオンにする

2. 乗算クォータニオン

$$\mathbf{q}\mathbf{v}\mathbf{q}^*$$

単位ベクトル (v_x, v_y, v_z) 周りの θ ラジアン回転を現すクォータニオン

$$\begin{pmatrix} \cos\frac{\theta}{2} \\ v_x \sin\frac{\theta}{2} \\ v_y \sin\frac{\theta}{2} \\ v_z \sin\frac{\theta}{2} \end{pmatrix}$$

(2) の回転クォータニオンは次で作成する (中々手ごわい)

次の技を駆使すると綺麗な回転になる

(a) unit 化

(b) 左右からかける

(c) 半分の角度 $\frac{\theta}{2}$

(d) 共役 4 元数

3. \mathbf{q} 回転角と回転軸の 2 要素を持つクォータニオンの作成

- 回転軸を選ぶ (x,y,z)
- 回転角はラジアンで与える
- 度数のラジアンへの変換には `rfd(numeric.ijs` に入っている) を用いる

```
clean ,.(cos;sin) rfd 30 45 60 90
```

```
+-----+
```

```
|0.866025 0.707107 0.5 0| NB. cos
```

```
+-----+
```

```
|0.5 0.707107 0.866025 1| NB. sin
```

```
+-----+
30      45      60      90 NB. degree
```

- z 軸で 90° 回す

```
mk_q 90;    0 0 1
0.707107 0 0 0.707107
```

- x,y,x 軸で 120° 回す。マイナス指定 OK

```
mk_q 120;1 _1 _1
0.5 0.5 _0.5 _0.5
```

4. 回転公式

$$q' = qvq^*$$

$$q = q_0 + \mathbf{q} = \cos\theta + \mathbf{u}\sin\theta$$

$$\mathbf{u} = \frac{\mathbf{q}}{|\mathbf{q}|}$$

$$qvq^* = (q_0^2 - |\mathbf{q}|^2)\mathbf{v} + 2(\mathbf{q} \cdot \mathbf{v})\mathbf{q} + 2q_0(\mathbf{q} \times \mathbf{v})$$

5. コメント

- 絶対値 1 の複素数とクォータニオンは回転を現す
- 共役は逆回転になる。ω 軸を含む逆回転で余計な回転を相殺する
- ω 軸を含まない回転は 2 倍回転することになるので $\frac{1}{2}$ しておく
- ω は回転に影響しないが、あればクォータニオンになるので計算が楽な 0 にしておく。

6. 回転クォータニオン作成スクリプト

Degree 用

```
mk_q=:3 : 0
'degree axis0'=. y
theta=: rfd degree
axis=: unitQ axis0
a0=: (cos;sin) -: theta
clean (>{. a0), (>{: a0) * }. axis
)
```

ラディアン用

```
mk_q_rad=:3 : 0
'theta axis0'=. y
axis=: unitQ axis0
a0=: (cos;sin) -: theta
clean (>{. a0), (>{: a0) * }. axis
)
```

7. 解説

- xyz 軸は最初にユニット化

- 先に-: で $\frac{\theta}{2}$ してある
- output はこのままで unit 化されている?
- clean は numeric.ijs に入っている数値計算の塵取り関数

5.2 計算例

回転軸と回転角は複雑な表現もあるので注意しなければならない。

$$\frac{\sqrt{2}}{2}, \frac{\sqrt{2}}{2}k$$

```
mk_q 45;0 0 1
0.92388 0 0 0.382683
```

入力フォーマット

%b

```
] qp0=. (45;0 0 1) compose_qp 2 0 0
0.92388 0
      0 2
      0 0
0.382683 0
```

$$p' = 0, \sqrt{2}i + \sqrt{2}j$$

%c

```
rotateQ qp0
0 1.41421 1.41421 0
```

EX.2 次の例は axis の作成法が特殊である。

```
point:(1 -1 2)
θ: 60°
axis:yz plane
y 軸に 60 度 - cos 60 度, z 軸は sin 60 度
```

- $p = i - j + 2k$
- yx plane は jk
- $\cos 60^\circ j + \sin 60^\circ k = \frac{1}{2}j + \frac{\sqrt{3}}{2}k$

%d

```

(cos;sin) rfd 60
+----+-----+
|0.5|0.866025|
+----+-----+
  -: %:3
0.866025
• q =  $\frac{\sqrt{3}}{2} + \frac{1}{2}\mathbf{u} = \frac{\sqrt{3}}{2} + \frac{1}{4}\mathbf{j} + \frac{\sqrt{3}}{4}\mathbf{k}$ 
• q* =  $\frac{\sqrt{3}}{2} - \frac{1}{4}\mathbf{j} - \frac{\sqrt{3}}{4}\mathbf{k}$ 
      qp =  $\frac{1}{4}((1 - 2\sqrt{3}) + (2 + 3\sqrt{3})\mathbf{i} - \sqrt{3}\mathbf{j} + (4\sqrt{3} - 1)\mathbf{k})$ 
      qpq* =  $\frac{1}{8}((10 - 4\sqrt{3}) + (2 + 3\sqrt{3})\mathbf{i} + (1 + 2\sqrt{3})\mathbf{j} + (14 - 3\sqrt{3})\mathbf{k})$ 

 $\frac{\sqrt{3}}{2}$ 
%e
] a=. mk_q 60; 0,>(cos;sin) rfd 60
0.866025 0 0.25 0.433013

]qp=. a,.0,1 _1 2
0.866025 0
      0 1
      0.25 _1
0.433013 2

%f
multipleQ qp
_0.616025 1.79904 _0.433013 1.48205
OK      OK      ?      OK      NB. EX. miss?

%g
rotateQ qp
0 2.11603 0.558013 1.10048
OK OK OK OK

```

6 マトリクスを用いた掛け算

クォータニオンの機能をそのまま Matrix form に移すものである。

クォータニオンはマトリクスとなじみが良い。マトリクスへ組み上げる構成は文献から探して拝借した。

*1

6.1 multipleQ の乗算マトリクスの作成

次のようなクォータニオンの係数をマトリクスにくみ上げると乗算クォータニオンでの掛け算ができる。

1. 乗算マトリクスの式

$$\begin{pmatrix} q_\omega & -q_z & q_y & q_x \\ q_z & q_\omega & -q_x & q_y \\ q_y & q_x & q_\omega & q_z \\ -q_x & -q_y & -q_z & -q_\omega \end{pmatrix} \begin{pmatrix} p_x \\ p_y \\ p_z \\ p_\omega \end{pmatrix}$$

このマトリクスは ω が最後に来る。

マイナスの作用はキャンセル

2. 計算とスクリプト

*1 時には転記ミスも見受けられるが、規則性があるので理解できる

```

%a
mk_Qmat_sub =: 3 : 0
NB. Usage: u qp // q * p
NB. for qp
NB. 'q0 p0'=: { |: y NB. qp
q0=. y
ind0=: ;(0 3 2 1);(3 0 1 2);(2 1 0 3);1 2 3 0
ind1=: (1 _1 1 1), (1 1 _1 1),(_1 1 1 1),_1 _1 _1 1
mat0=: 4 4 $ ind1 * ind0{q0
)

%b
mk_Qmat_sub {."1 qp
3 _1 _2 1
1 3 _1 _2
2 1 3 1
_1 2 _1 3

multipleQ_mat=: 3 : 0
'q0 p0'=: { |: y NB. qp
mat0=: mk_Qmat_sub q0
p1=: 1 |. p0
_1 |. mat0 +/ . * p1
)

%c
multipleQ_mat qp
8 _9 _2 11

```

6.2 rotareQ の回転マトリクス判の作成

クォータニオンを回転マトリクスに組み上げると共役クォータニオンを右からかける計算は不要となる

クォータニオン形式の 4×4 のマトリクスの文献は多くない。
このマトリクスも ω が最後となっているがそのままとした。

1. 回転マトリクスの式

$$Q(x, y, z, \omega) = (v \cdot \sin \frac{\theta}{2}, v \cdot \cos \frac{\theta}{2})$$

$$\begin{pmatrix} 1 - 2(y^2 + z^2) & 2(xy - z\omega) & 2(xz + y\omega) & 0 \\ 2(xy + z\omega) & 1 - 2(x^2 + z^2) & 2(yz - x\omega) & 0 \\ 2(xz - y\omega) & 2(yz + x\omega) & 1 - 2(x^2 + y^2) & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

こちらも目を凝らすと、同じ虚数単位の積と、異なる虚数単位の積に区別できる

2. 計算

(a) EX.1

```

] qp0=. (45;0 0 1) compose_qp 2 0 0
0.92388 0

```



```

0 2
0 0
0.382683 0

```

```

mk_Q_rotmat {"1 qp0
0.707107 _0.707107 0 0
0.707107 0.707107 0 0
0 0 1 0
0 0 0 1

```

```

rotateQ_rotmat qp0
0 1.41421 1.41421 0
 $\sqrt{2}i + \sqrt{2}j$ 

```

(b) EX.2

```

qp
0.866025 0
0 1
0.25 _1
0.433013 2

```

```

mk_Q_rotmat qp
0.5 _2 0.875 0
_9 _9 4 0
0.75 _3 0.625 0
0 0 0 1
rotateQ qp
0 2.11603 0.558013 1.10048

```

```

rotateQ_rotmat qp
0 2.11603 0.558013 1.10048

```

3. Script

```

rotateQ_rotmat=: 3 : 0
'q0 p0'=:{|: y
mat=: mk_Q_rotmat q0
p1=. 1 |. p0
_1 |. mat +/ . * p1
)

mk_Q_rotmat=: 3 : 0
NB. y is q0
NB. order of mat is i ,j, k, omega
b0=(2 3;1 3;1 2){ L:0 *: y
b01=: ; -. @ +: @ +/ L:0 b0
index0=: { ~. /:~("1) 2{."1 tap 4
a0=:index0 { L:0 y
a01=: */ L:0 a0
a02=: _2<\;(3 2 4 1 5 0){a01
a03=:a02, 1 _1 * L:0 a02
a04=: +: ; +/ L:0 a03
tmp0=: b01,a04
tmp1=:3 3 $ 0 6 4 3 1 8 7 5 2 { tmp0
(tmp1,0 0 0),. 0 0 0 1
)

```

- tap: Vocablaly に入っている A.(辞書式順序) を用いた組み合わせのアイデア

```

index0
+---+---+---+---+---+---+
|0 1|0 2|0 3|1 2|1 3|2 3|
+---+---+---+---+---+---+

```

- 先に組み合わせを計算してから順に並べており、内積、外積は使っていない

7 公道走行

ツールや言語のライブラリに依らず一から組み上げたスクリプトなのでテスト走行を試みよう。

7.1 仮免許

頭書の例題をマニュアルモードで試してみよう。

1. 高校数学の美しい物語

- q


```
mk_q 180;1 1 0
0 0.707107 0.707107 0
```
- 入力の組み上げ


```
( 180;1 1 0) compose_qp 3 0 0
0 0
```

```
0.707107 3
0.707107 0
      0 0
```

- $q \cdot p$

```
multipleQ_mat a
_2.12132 0 0 _2.12132
```

- qpq^*

```
rotateQ_mat a
0 0 3 0
```

2. 物理のカギしっぽ

- mk_q

```
mk_q 90;0 1 0
0.707107 0 0.707107 0
```

- $compose_qp$

```
(90;0 1 0) compose_qp 0 0 1
```

```
0.707107 0
      0 0
0.707107 0
      0 1
```

- $calc$

```
rotateQ_mat a
0 1 0 0
```

2 問とも正解!

7.2 初心者免許

双葉マーク付きの運転免許が取れた。

Ex10 . <https://wisteriahill.sakura.ne.jp/CMS/WordPress/2018/10/10/quaternion-memo/>

点 $A(3,0,0)$ を $y = x, z = 0$ で表される直線の
周りを 180° 回転させた点 B の座標を求める

```

    mk_q 180;1 1 0
0 0.707107 0.707107 0

    clean rotateQ_rotmat qp0
0 0 3 0

] qp0=. (180;1 1 0) compose_qp 3 0 0 rotateQ qp0
    0 0
0.707107 3
0.707107 0
    0 0

ans = 3j = 0, 3, 0

```

8 Conclusion

次の Script の実戦練習も終えた。不足分は追って追加しよう。

入力	mk_q	compose_qp
	mk_q_rad	
掛け算	multipleQ	multipleQ_mat
回転	rotateQ	rotateQ_rotmat

クォータニオンは素直な数であり、丁寧にスクリプトを作成すればクォータニオンの数学を自由に取り扱うことができる。

掛け算と回転のスクリプトはチキチキバンバン方式とマトリクス方式の二通り作成した。差異はないので好みで使い分けることができる。

References

クォータニオンの回転マトリクス

<https://medium.com/@behreajj/3d-rotations-in-processing-vectors-matrices-quaternions-10e2fed5f0a3>

用語のメモ

- pure quaternian $\omega = 0$ のクォータニオン
- ジンバルロック: 中華料理店の回転テーブルで、真中に醤油を置くと回しても取れない状態